A *Mathematica* Tutorial 31 January 2007

1. Plots and Integrals

```
Plot[x^3 / (Exp[x] - 1), {x, 0, 10}]
```

```
Integrate[x^3 / (Exp[x] - 1), {x, 0, 10}]
```

Enclose the preceding instruction in N[ ] and execute it again. Then delete the brackets to use "NIntegrate".

```
planck = x^3 / (Exp[x] - 1)
```

```
NIntegrate[planck, {x, 0, Infinity}]
```

Delete the initial N to get an exact answer.

```
Plot[3 t^3 NIntegrate[x^4 Exp[x] / (Exp[x] - 1)^2, {x, 0, 1 / t}], {t, 0, 2}]
```

Tips: * Hit shift-return (or keypad enter) to send input to *Mathematica*.
* Be careful with capitalization and types of brackets.
* To abort a calculation, use the Abort command under the Kernel menu.
* To collapse and expand groups of cells, double-click on the enclosing brackets at the right margin.
* To add a comment cell like this, click where you want it and before you type anything, choose the desired style from the Format menu.

2. Root Finding, Minimization, Sums, Derivatives

```
Plot[{Tanh[2 x], x}, {x, -2, 2}]
```

```
solution = FindRoot[Tanh[2 x] == x, {x, 1}]
```

```
Tanh[2 x] /. solution
```

```
?FindRoot
```

```
FindMaximum[x^3 / (Exp[x] - 1), {x, 3}]
```

```
deriv = D[planck, x]
```

```
FindRoot[deriv == 0, {x, 3}]
```

```
Sum[1 / n^2, {n, 1, 10}]
```

Change upper limit from 10 to Infinity. Then change exponent "2" to "s".

```
Plot[Sum[Sin[k * x] / k, {k, 1, 25, 2}], {x, 0, 2 Pi}]
```

3. Lists

```
factorials = Table[n!, {n, 1, 10}]
```

```
factorials[[5]]
```

```
stirling = Table[n^n * Exp[-n] Sqrt[2 Pi * n], {n, 1, 10}]
```

Enclose the Table function above in N[ ].

```
fracError = (factorials - stirling) / factorials
```

```
pairs = Table[{n, fracError[[n]]}, {n, 1, 10}]
```

```
MatrixForm[pairs]
```

```
pairs[[5, 2]]
```

```
ListPlot[pairs]
```

```
ListPlot[pairs, Frame → True, PlotRange → {{0, 11}, {0, 0.09}},
 PlotStyle → {Hue[.75], AbsolutePointSize[5]}]
```

(Type hyphen greater-than, "->", to get the arrow symbol. for assigning values.)

```
Options[ListPlot]
```

```
? Frame
```

4. Reading and Writing Data

```
Directory[]
```

```
SetDirectory["~/Desktop"]
```

```
Export["pairs.txt", pairs, "Table"]
```

Point a web browser to physics.weber.edu/schroeder/mma/data.txt, then save to the desktop as "data.txt".

```
importedData = Import["data.txt", "Table"];
```

```
TableForm[importedData]
```

```
dataPlot = ListPlot[importedData, PlotStyle → Hue[.7]];
```

```
curvePlot = Plot[7.55 - 0.0048 x + 2.14 * Exp[-(x / 18) ^2], {x, -100, 100},
   PlotStyle → {AbsoluteThickness[2], Hue[0]}, DisplayFunction → Identity];
Show[dataPlot, curvePlot];
```

(In this and all subsequent plots, try removing the various options to see what effects they have.)

5. More Graphics

```
ParametricPlot[{Sin[2 t], Sin[3 t]}, {t, 0, 2 Pi},
  AspectRatio → Automatic, GridLines → Automatic, Axes → False,
  Frame → True, PlotStyle → {Hue[0], AbsoluteThickness[2]}];

a = 1; epsilon = .9;
r = a (1 - epsilon^2) / (1 + epsilon * Cos[theta]);
ParametricPlot[{r * Cos[theta], r * Sin[theta]}, {theta, 0, 2 Pi},
  AspectRatio → Automatic, PlotStyle → {Hue[.75], AbsoluteThickness[2]}];

h3d = (2 z^2 - x^2)^2 Exp[-2 Sqrt[x^2 + z^2] / 3]

DensityPlot[h3d, {x, -20, 20}, {z, -20, 20},
  Mesh → False, PlotPoints → 500, PlotRange → All, ColorFunction → Hue ];

ContourPlot[h3d, {x, -20, 20}, {z, -20, 20},
  PlotPoints → 500, PlotRange → All, ContourShading → False, Contours → 12];

Plot3D[h3d, {x, -20, 20}, {z, -20, 20}, PlotRange → All,
  Mesh → False, PlotPoints → 200, Boxed → False, Axes → False];

k = 10;
wavepacket = Exp[-k^2 / 4] * Exp[(I * x + k / 2)^2 / (1 + 2 I * t)] / Sqrt[1 + 2 I * t]

Table[Plot[{Re[wavepacket], Im[wavepacket]}, {x, -2, 14},
   PlotPoints → 500, PlotRange → {-1, 1}, PlotStyle → {{Hue[0]}, {Hue[.6]}},
   AspectRatio → .3, Ticks → None, ImageSize → 600], {t, 0, 1, .01}];
```

6. Symbolic Operations

```
Clear[epsilon, k]

z = 1 + Exp[-beta * epsilon]

energy = -(1 / z) D[z, beta]

energy = Simplify[-(1 / z) D[z, beta]] /. beta → 1 / (k t)

heatcap = D[energy, t]

Plot[heatcap /. {epsilon → 1, k → 1}, {t, 0, 3}];
```

7. Differential Equations

```
solution = NDSolve[
   {theta''[t] == -Sin[theta[t]], theta[0] == .1 Pi, theta'[0] == 0}, theta, {t, 0, 50}];
Plot[theta[t] /. solution, {t, 0, 50}, PlotStyle → {AbsoluteThickness[2], Hue[0]}];

FindRoot[(theta[t] /. solution) == 0, {t, 5}]

e = 1.5; xmax = 5;
solution = NDSolve[ { psi''[x] == (-2 e + x^2) psi[x],
    psi[-xmax] == 0, psi'[-xmax] == 0.001 }, psi, {x, -xmax, xmax}];
Plot[ psi[x] /. solution , {x, -xmax, xmax}];
```